// File: @openzeppelin/contracts/token/ERC20/IERC20.sol

// SPDX-License-Identifier: MIT

pragma solidity >=0.6.0 <0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `recipient`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address recipient, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `sender` to `recipient` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(address sender, address recipient, uint256 amount) external returns (bool);

    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);
}

// File: @openzeppelin/contracts/GSN/Context.sol

/*
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with GSN meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address payable) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes memory) {
        this; // silence state mutability warning without generating bytecode - see https://github.com/ethereum/solidity/issues/2691
        return msg.data;
    }
}

// File: @openzeppelin/contracts/math/SafeMath.sol

/**
 * @dev Wrappers over Solidity's arithmetic operations with added overflow
 * checks.
 *
 * Arithmetic operations in Solidity wrap on overflow. This can easily result
 * in bugs, because programmers usually assume that an overflow raises an
 * error, which is the standard behavior in high level programming languages.
 * `SafeMath` restores this intuition by reverting the transaction when an
 * operation overflows.
 *
 * Using this library instead of the unchecked operations eliminates an entire
 * class of bugs, so it's recommended to use it always.
 */
library SafeMath {
    /**
     * @dev Returns the addition of two unsigned integers, reverting on
     * overflow.
     *
     * Counterpart to Solidity's `+` operator.
     *
     * Requirements:
     *
     * - Addition cannot overflow.
     */
    function add(uint256 a, uint256 b) internal pure returns (uint256) {
        uint256 c = a + b;
        require(c >= a, "SafeMath: addition overflow");

        return c;
    }

    /**
     * @dev Returns the subtraction of two unsigned integers, reverting on
     * overflow (when the result is negative).
     *
     * Counterpart to Solidity's `-` operator.
     *
     * Requirements:
     *
     * - Subtraction cannot overflow.
     */
    function sub(uint256 a, uint256 b) internal pure returns (uint256) {
        return sub(a, b, "SafeMath: subtraction overflow");
    }

    /**
     * @dev Returns the subtraction of two unsigned integers, reverting with custom message on
     * overflow (when the result is negative).
     *
     * Counterpart to Solidity's `-` operator.
     *
     * Requirements:
     *
     * - Subtraction cannot overflow.
     */
    function sub(uint256 a, uint256 b, string memory errorMessage) internal pure returns (uint256) {
        require(b <= a, errorMessage);
        uint256 c = a - b;

        return c;
    }

    /**
     * @dev Returns the multiplication of two unsigned integers, reverting on
     * overflow.
     *
     * Counterpart to Solidity's `*` operator.
     *
     * Requirements:
     *
     * - Multiplication cannot overflow.
     */
    function mul(uint256 a, uint256 b) internal pure returns (uint256) {
        // Gas optimization: this is cheaper than requiring 'a' not being zero, but the
        // benefit is lost if 'b' is also tested.
        // See: https://github.com/OpenZeppelin/openzeppelin-contracts/pull/522
        if (a == 0) {
            return 0;
        }

        uint256 c = a * b;
        require(c / a == b, "SafeMath: multiplication overflow");

        return c;
    }

    /**
     * @dev Returns the integer division of two unsigned integers. Reverts on
     * division by zero. The result is rounded towards zero.
     *
     * Counterpart to Solidity's `/` operator. Note: this function uses a
     * `revert` opcode (which leaves remaining gas untouched) while Solidity
     * uses an invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function div(uint256 a, uint256 b) internal pure returns (uint256) {
        return div(a, b, "SafeMath: division by zero");
    }

    /**
     * @dev Returns the integer division of two unsigned integers. Reverts with custom message on
     * division by zero. The result is rounded towards zero.
     *
     * Counterpart to Solidity's `/` operator. Note: this function uses a
     * `revert` opcode (which leaves remaining gas untouched) while Solidity
     * uses an invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function div(uint256 a, uint256 b, string memory errorMessage) internal pure returns (uint256) {
        require(b > 0, errorMessage);
        uint256 c = a / b;
        // assert(a == b * c + a % b); // There is no case in which this doesn't hold

        return c;
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers. (unsigned integer modulo),
     * Reverts when dividing by zero.
     *
     * Counterpart to Solidity's `%` operator. This function uses a `revert`
     * opcode (which leaves remaining gas untouched) while Solidity uses an
     * invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function mod(uint256 a, uint256 b) internal pure returns (uint256) {
        return mod(a, b, "SafeMath: modulo by zero");
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers. (unsigned integer modulo),
     * Reverts with custom message when dividing by zero.
     *
     * Counterpart to Solidity's `%` operator. This function uses a `revert`
     * opcode (which leaves remaining gas untouched) while Solidity uses an
     * invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function mod(uint256 a, uint256 b, string memory errorMessage) internal pure returns (uint256) {
        require(b != 0, errorMessage);
        return a % b;
    }
}

// File: @openzeppelin/contracts/token/ERC20/ERC20.sol




/**
 * @dev Implementation of the {IERC20} interface.
 *
 * This implementation is agnostic to the way tokens are created. This means
 * that a supply mechanism has to be added in a derived contract using {_mint}.
 * For a generic mechanism see {ERC20PresetMinterPauser}.
 *
 * TIP: For a detailed writeup see our guide
 * https://forum.zeppelin.solutions/t/how-to-implement-erc20-supply-mechanisms/226[How
 * to implement supply mechanisms].
 *
 * We have followed general OpenZeppelin guidelines: functions revert instead
 * of returning `false` on failure. This behavior is nonetheless conventional
 * and does not conflict with the expectations of ERC20 applications.
 *
 * Additionally, an {Approval} event is emitted on calls to {transferFrom}.
 * This allows applications to reconstruct the allowance for all accounts just
 * by listening to said events. Other implementations of the EIP may not emit
 * these events, as it isn't required by the specification.
 *
 * Finally, the non-standard {decreaseAllowance} and {increaseAllowance}
 * functions have been added to mitigate the well-known issues around setting
 * allowances. See {IERC20-approve}.
 */
contract ERC20 is Context, IERC20 {
    using SafeMath for uint256;

    mapping (address => uint256) private _balances;

    mapping (address => mapping (address => uint256)) private _allowances;

    uint256 private _totalSupply;

    string private _name;
    string private _symbol;
    uint8 private _decimals;

    /**
     * @dev Sets the values for {name} and {symbol}, initializes {decimals} with
     * a default value of 18.
     *
     * To select a different value for {decimals}, use {_setupDecimals}.
     *
     * All three of these values are immutable: they can only be set once during
     * construction.
     */
    constructor (string memory name_, string memory symbol_) public {
        _name = name_;
        _symbol = symbol_;
        _decimals = 18;
    }

    /**
     * @dev Returns the name of the token.
     */
    function name() public view returns (string memory) {
        return _name;
    }

    /**
     * @dev Returns the symbol of the token, usually a shorter version of the
     * name.
     */
    function symbol() public view returns (string memory) {
        return _symbol;
    }

    /**
     * @dev Returns the number of decimals used to get its user representation.
     * For example, if `decimals` equals `2`, a balance of `505` tokens should
     * be displayed to a user as `5,05` (`505 / 10 ** 2`).
     *
     * Tokens usually opt for a value of 18, imitating the relationship between
     * Ether and Wei. This is the value {ERC20} uses, unless {_setupDecimals} is
     * called.
     *
     * NOTE: This information is only used for _display_ purposes: it in
     * no way affects any of the arithmetic of the contract, including
     * {IERC20-balanceOf} and {IERC20-transfer}.
     */
    function decimals() public view returns (uint8) {
        return _decimals;
    }

    /**
     * @dev See {IERC20-totalSupply}.
     */
    function totalSupply() public view override returns (uint256) {
        return _totalSupply;
    }

    /**
     * @dev See {IERC20-balanceOf}.
     */
    function balanceOf(address account) public view override returns (uint256) {
        return _balances[account];
    }

    /**
     * @dev See {IERC20-transfer}.
     *
     * Requirements:
     *
     * - `recipient` cannot be the zero address.
     * - the caller must have a balance of at least `amount`.
     */
    function transfer(address recipient, uint256 amount) public virtual override returns (bool) {
        _transfer(_msgSender(), recipient, amount);
        return true;
    }

    /**
     * @dev See {IERC20-allowance}.
     */
    function allowance(address owner, address spender) public view virtual override returns (uint256) {
        return _allowances[owner][spender];
    }

    /**
     * @dev See {IERC20-approve}.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     */
    function approve(address spender, uint256 amount) public virtual override returns (bool) {
        _approve(_msgSender(), spender, amount);
        return true;
    }

    /**
     * @dev See {IERC20-transferFrom}.
     *
     * Emits an {Approval} event indicating the updated allowance. This is not
     * required by the EIP. See the note at the beginning of {ERC20}.
     *
     * Requirements:
     *
     * - `sender` and `recipient` cannot be the zero address.
     * - `sender` must have a balance of at least `amount`.
     * - the caller must have allowance for ``sender``'s tokens of at least
     * `amount`.
     */
    function transferFrom(address sender, address recipient, uint256 amount) public virtual override returns (bool) {
        _transfer(sender, recipient, amount);
        _approve(sender, _msgSender(), _allowances[sender][_msgSender()].sub(amount, "ERC20: transfer amount exceeds allowance"));
        return true;
    }

    /**
     * @dev Atomically increases the allowance granted to `spender` by the caller.
     *
     * This is an alternative to {approve} that can be used as a mitigation for
     * problems described in {IERC20-approve}.
     *
     * Emits an {Approval} event indicating the updated allowance.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     */
    function increaseAllowance(address spender, uint256 addedValue) public virtual returns (bool) {
        _approve(_msgSender(), spender, _allowances[_msgSender()][spender].add(addedValue));
        return true;
    }

    /**
     * @dev Atomically decreases the allowance granted to `spender` by the caller.
     *
     * This is an alternative to {approve} that can be used as a mitigation for
     * problems described in {IERC20-approve}.
     *
     * Emits an {Approval} event indicating the updated allowance.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     * - `spender` must have allowance for the caller of at least
     * `subtractedValue`.
     */
    function decreaseAllowance(address spender, uint256 subtractedValue) public virtual returns (bool) {
        _approve(_msgSender(), spender, _allowances[_msgSender()][spender].sub(subtractedValue, "ERC20: decreased allowance below zero"));
        return true;
    }

    /**
     * @dev Moves tokens `amount` from `sender` to `recipient`.
     *
     * This is internal function is equivalent to {transfer}, and can be used to
     * e.g. implement automatic token fees, slashing mechanisms, etc.
     *
     * Emits a {Transfer} event.
     *
     * Requirements:
     *
     * - `sender` cannot be the zero address.
     * - `recipient` cannot be the zero address.
     * - `sender` must have a balance of at least `amount`.
     */
    function _transfer(address sender, address recipient, uint256 amount) internal virtual {
        require(sender != address(0), "ERC20: transfer from the zero address");
        require(recipient != address(0), "ERC20: transfer to the zero address");

        _beforeTokenTransfer(sender, recipient, amount);

        _balances[sender] = _balances[sender].sub(amount, "ERC20: transfer amount exceeds balance");
        _balances[recipient] = _balances[recipient].add(amount);
        emit Transfer(sender, recipient, amount);
    }

    /** @dev Creates `amount` tokens and assigns them to `account`, increasing
     * the total supply.
     *
     * Emits a {Transfer} event with `from` set to the zero address.
     *
     * Requirements:
     *
     * - `to` cannot be the zero address.
     */
    function _mint(address account, uint256 amount) internal virtual {
        require(account != address(0), "ERC20: mint to the zero address");

        _beforeTokenTransfer(address(0), account, amount);

        _totalSupply = _totalSupply.add(amount);
        _balances[account] = _balances[account].add(amount);
        emit Transfer(address(0), account, amount);
    }

    /**
     * @dev Destroys `amount` tokens from `account`, reducing the
     * total supply.
     *
     * Emits a {Transfer} event with `to` set to the zero address.
     *
     * Requirements:
     *
     * - `account` cannot be the zero address.
     * - `account` must have at least `amount` tokens.
     */
    function _burn(address account, uint256 amount) internal virtual {
        require(account != address(0), "ERC20: burn from the zero address");

        _beforeTokenTransfer(account, address(0), amount);

        _balances[account] = _balances[account].sub(amount, "ERC20: burn amount exceeds balance");
        _totalSupply = _totalSupply.sub(amount);
        emit Transfer(account, address(0), amount);
    }

    /**
     * @dev Sets `amount` as the allowance of `spender` over the `owner` s tokens.
     *
     * This internal function is equivalent to `approve`, and can be used to
     * e.g. set automatic allowances for certain subsystems, etc.
     *
     * Emits an {Approval} event.
     *
     * Requirements:
     *
     * - `owner` cannot be the zero address.
     * - `spender` cannot be the zero address.
     */
    function _approve(address owner, address spender, uint256 amount) internal virtual {
        require(owner != address(0), "ERC20: approve from the zero address");
        require(spender != address(0), "ERC20: approve to the zero address");

        _allowances[owner][spender] = amount;
        emit Approval(owner, spender, amount);
    }

    /**
     * @dev Sets {decimals} to a value other than the default one of 18.
     *
     * WARNING: This function should only be called from the constructor. Most
     * applications that interact with token contracts will not expect
     * {decimals} to ever change, and may work incorrectly if it does.
     */
    function _setupDecimals(uint8 decimals_) internal {
        _decimals = decimals_;
    }

    /**
     * @dev Hook that is called before any transfer of tokens. This includes
     * minting and burning.
     *
     * Calling conditions:
     *
     * - when `from` and `to` are both non-zero, `amount` of ``from``'s tokens
     * will be to transferred to `to`.
     * - when `from` is zero, `amount` tokens will be minted for `to`.
     * - when `to` is zero, `amount` of ``from``'s tokens will be burned.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _beforeTokenTransfer(address from, address to, uint256 amount) internal virtual { }
}

// File: @openzeppelin/contracts/token/ERC20/ERC20Burnable.sol



/**
 * @dev Extension of {ERC20} that allows token holders to destroy both their own
 * tokens and those that they have an allowance for, in a way that can be
 * recognized off-chain (via event analysis).
 */
abstract contract ERC20Burnable is Context, ERC20 {
    using SafeMath for uint256;

    /**
     * @dev Destroys `amount` tokens from the caller.
     *
     * See {ERC20-_burn}.
     */
    function burn(uint256 amount) public virtual {
        _burn(_msgSender(), amount);
    }

    /**
     * @dev Destroys `amount` tokens from `account`, deducting from the caller's
     * allowance.
     *
     * See {ERC20-_burn} and {ERC20-allowance}.
     *
     * Requirements:
     *
     * - the caller must have allowance for ``accounts``'s tokens of at least
     * `amount`.
     */
    function burnFrom(address account, uint256 amount) public virtual {
        uint256 decreasedAllowance = allowance(account, _msgSender()).sub(amount, "ERC20: burn amount exceeds allowance");

        _approve(account, _msgSender(), decreasedAllowance);
        _burn(account, amount);
    }
}

// File: contracts/ChangeNowSimpleStaking.sol



//import "@openzeppelin/contracts/utils/Pausable.sol";


contract ChangeNowSimpleStaking is ERC20Burnable /*, Pausable*/ {
    using SafeMath for uint256;

    uint256 constant MIN_EQUITY = 1000;

    address payable private _owner;
    address payable private _successor;

    address public token;

    uint256 public minStake;
    uint256 public maxStake;
    uint256 public freezeDuration;

    mapping(address => uint256) private _stakeAt;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);
    event MinMaxStakeChanged(uint256 newMinStake, uint256 newMaxStake);
    event FreezeDurationChanged(uint256 newFreezeDuration);

    constructor(
        address _token,
        uint256 _minStake,
        uint256 _maxStake,
        uint256 _freezeDuration,
        string memory _name,
        string memory _symbol
    )
        ERC20(_name, _symbol)
    {
        _setupDecimals(ERC20(_token).decimals());

        token = _token;
        _setMinMaxStake(_minStake, _maxStake);
        _setFreezeDuration(_freezeDuration);

        _owner = msg.sender;
        emit OwnershipTransferred(address(0), msg.sender);
    }


    /**
     * @notice Checks that caller is owner.
     */
    modifier onlyOwner() {
        require(msg.sender == _owner, "Caller is not owner");
        _;
    }


    /**
     * @notice Checks that caller is not frozen for the current time
     */
    modifier notFrozen() {
        uint256 lastStakeAt = _stakeAt[msg.sender];
        if (lastStakeAt > 0) {
            require(block.timestamp - lastStakeAt > freezeDuration, 'frozen');
        }
        _;
    }


    /**
     * @notice Sets new successor (only owner).
     */
    function setSuccessor(address payable newSuccessor)
        external
        onlyOwner
    {
        _successor = newSuccessor;
    }


    /**
     * @notice Takes ownership (only successor).
     */
    function takeOwnership()
        external
    {
        require(msg.sender == _successor, "Caller is not successor");

        address previousOwner = _owner;
        _owner = msg.sender;
        _successor = address(0);

        emit OwnershipTransferred(previousOwner, msg.sender);
    }


    /**
     * @notice Sets new min & max stake amounts (only owner).
     */
    function setMinMaxStake(uint256 _minStake, uint256 _maxStake)
        external
        onlyOwner
    {
        _setMinMaxStake(_minStake, _maxStake);
    }


    /**
     * @notice Sets new freeze duration
     */
    function setFreezeDuration(uint256 _freezeDuration)
        external
        onlyOwner
    {
        _setFreezeDuration(_freezeDuration);
    }


    /**
     * @notice Pauses the staking, but not redeeming (only owner).
     */
/***************************
    function pause()
        external
        onlyOwner
    {
        _pause();
    }
***************************/

    /**
     * @notice Unpauses paused staking (only owner).
     */
/***************************
    function unpause()
        external
        onlyOwner
    {
        _unpause();
    }
***************************/


    /**
     * @notice Destroys the contract (when empty).
     */
    function destroy() external onlyOwner {
        require(totalSupply() == 0, 'has holders');

        // is there any token balance (sent after last redeeming)?
        // send whole balance to the owner
        uint256 totalTokenBalance = IERC20(token).balanceOf(address(this));
        if (totalTokenBalance > 0) {
            IERC20(token).transfer(_owner, totalTokenBalance);
        }

        // destroy
        selfdestruct(_owner);
    }


    /**
     * @notice Shows underlying token balance of an account.
     */
    function balanceOfUnderlying(address account)
        external
        view
        returns (uint256)
    {
        return _equityToToken(balanceOf(account));
    }


    function tokenToEquity(uint256 tokenAmount)
        external
        view
        returns (uint256)
    {
        return _tokenToEquity(tokenAmount);
    }


    function equityToToken(uint256 equity)
        external
        view
        returns (uint256)
    {
        return _equityToToken(equity);
    }


    function frozenTill(address account)
        external
        view
        returns (uint256)
    {
        uint256 lastStakeAt = _stakeAt[account];
        return lastStakeAt > 0 ? lastStakeAt + freezeDuration : 0;
    }


    /**
     * @notice Stakes specified token amounts to the staking contract.
     * The tokens must be approved to transfer from sender by the staking contract.
     */
    function stake(uint256 tokenAmount)
        external
        //whenNotPaused
        returns (uint256 equity)
    {
        uint256 allowance = IERC20(token).allowance(msg.sender, address(this));
        require(tokenAmount <= allowance, 'Not enough ERC20 allowance');

        uint256 accountTokenBalance = IERC20(token).balanceOf(msg.sender);
        require(tokenAmount <= accountTokenBalance, 'Not enough token balance');

        // get total token balance (common token pool) and total supply
        uint256 totalTokenBalance = IERC20(token).balanceOf(address(this));
        uint256 totalSupply = totalSupply();

        // check min / max stake
        uint256 accountStakingBalance =
            _equityToToken0(balanceOf(msg.sender), totalSupply, totalTokenBalance);
        require(tokenAmount >= minStake, 'Stake too small');
        require((tokenAmount.add(accountStakingBalance)) <= maxStake, 'Exceeded max stake');

        // calc equity
        if (totalSupply > 0) {
            equity = _tokenToEquity0(tokenAmount, totalTokenBalance, totalSupply);
            require(equity >= MIN_EQUITY, 'MIN_EQUITY: Stake too small');
        } else {
            equity = tokenAmount;
        }

        // transfer tokens
        require(
            IERC20(token).transferFrom(msg.sender, address(this), tokenAmount),
            'ERC20 transferFrom failed'
        );
        _mint(msg.sender, equity);

        // register last stake timestamp
        _stakeAt[msg.sender] = block.timestamp;
    }


    function redeem(uint256 equity)
        external
        notFrozen
        returns (uint256)
    {
        uint256 totalSupply = totalSupply();
        uint256 totalTokenBalance = IERC20(token).balanceOf(address(this));

        return _redeem(equity, totalSupply, totalTokenBalance);
    }


    function redeemAll()
        external
        notFrozen
        returns (uint256 equity, uint256 tokenAmount)
    {
        uint256 totalSupply = totalSupply();
        uint256 totalTokenBalance = IERC20(token).balanceOf(address(this));

        equity = balanceOf(msg.sender);
        tokenAmount = _equityToToken0(equity, totalSupply, totalTokenBalance);

        _redeem0(equity, tokenAmount);
    }


    function redeemUnderlying(uint256 tokenAmount)
        external
        notFrozen
        returns (uint256)
    {
        uint256 totalSupply = totalSupply();
        uint256 totalTokenBalance = IERC20(token).balanceOf(address(this));
        uint256 equity = _tokenToEquity0(tokenAmount, totalTokenBalance, totalSupply);

        return _redeem(equity, totalSupply, totalTokenBalance);
    }


    //
    //  Internal functions
    //

    function _setMinMaxStake(uint256 _minStake, uint256 _maxStake)
        private
    {
        require(_minStake <= _maxStake, 'minStake > maxStake');
        require(_minStake >= MIN_EQUITY, 'minStake < MIN_EQUITY');
        minStake = _minStake;
        maxStake = _maxStake;

        emit MinMaxStakeChanged(_minStake, _maxStake);
    }

    function _setFreezeDuration(uint256 _freezeDuration)
        private
    {
        freezeDuration = _freezeDuration;

        emit FreezeDurationChanged(_freezeDuration);
    }


    //
    //  Redeem internal functions
    //

    function _redeem0(uint256 equity, uint256 tokenAmount)
        private
    {
        if (equity > 0) {
            _burn(msg.sender, equity);
        }
        if (tokenAmount > 0) {
            require(IERC20(token).transfer(msg.sender, tokenAmount), 'ERC20 transfer failed');
        }
    }

    function _redeem(uint256 equity, uint256 totalSupply, uint256 totalTokenBalance)
        private
        returns (uint256)
    {
        // check MIN_EQUITY
        require(equity >= MIN_EQUITY, 'MIN_EQUITY: equity too small');

        // redeem only when funded
        require(totalSupply > 0 && totalTokenBalance > 0, 'No funds');

        // check account equity
        uint256 accountEquity = balanceOf(msg.sender);
        require(equity <= accountEquity, 'Not enough equity');

        // ensure that remaining equity is 0 or not less than MIN_EQUITY
        uint256 remainEquity = accountEquity - equity;
        require(
            remainEquity == 0 || remainEquity >= MIN_EQUITY,
            'MIN_EQUITY: remaining equity too small (hint: redeem all)'
        );

        // same for remaining underlying tokens
        uint256 remainTokenBalance =
            _equityToToken0(remainEquity, totalSupply, totalTokenBalance);
        require(
            remainTokenBalance == 0 || remainTokenBalance >= minStake,
            'MIN_STAKE: remaining stake balance too small (hint: redeem all)'
        );

        // calc underlying token amount
        uint256 tokenAmount =
            _equityToToken0(equity, totalSupply, totalTokenBalance);

        // redeem
        _redeem0(equity, tokenAmount);

        return tokenAmount;
    }


    //
    //  Helper internal functions
    //

    function _xxxToYyy_floor(uint256 xxxAmount, uint256 totalXxx, uint256 totalYyy)
        private pure
        returns (uint256)
    {
        if (totalXxx == 0) {
            return 0;
        }

        return xxxAmount.mul(totalYyy).div(totalXxx);
    }

    function _xxxToYyy_ceil(uint256 xxxAmount, uint256 totalXxx, uint256 totalYyy)
        private pure
        returns (uint256)
    {
        if (totalXxx == 0) {
            return 0;
        }

        uint256 nom = xxxAmount.mul(totalYyy).add(totalXxx).sub(1);
        uint256 quot = nom.div(totalXxx);

        return quot;
    }

    function _equityToToken0(uint256 equityAmount, uint256 totalEquity, uint256 totalTokenBalance)
        private
        pure
        returns (uint256)
    {
        return _xxxToYyy_floor(equityAmount, totalEquity, totalTokenBalance);
    }

    function _tokenToEquity0(uint256 tokenAmount, uint256 totalTokenBalance, uint256 totalEquity)
        private
        pure
        returns (uint256)
    {
        return _xxxToYyy_ceil(tokenAmount, totalTokenBalance, totalEquity);
    }

    function _tokenToEquity(uint256 tokenAmount)
        private
        view
        returns (uint256)
    {
        uint256 totalSupply = totalSupply();
        uint256 totalTokenBalance = IERC20(token).balanceOf(address(this));

        return _tokenToEquity0(tokenAmount, totalTokenBalance,totalSupply);
    }

    function _equityToToken(uint256 equity)
        private
        view
        returns (uint256)
    {
        uint256 totalSupply = totalSupply();
        uint256 totalTokenBalance = IERC20(token).balanceOf(address(this));

        return _equityToToken0(equity, totalSupply, totalTokenBalance);
    }
}